

# Jednoduchý model stability alternátoru


## Description

Jednoduchý model stability alternátoru s direktivní regulací buzení a turbíny.

## System Parameters

$P_{max1} = 225.03$	[MW]	Maximalni cinny vykon alternatoru v predporuchovem stavu
$P_{max2} = 105.23$	[MW]	Maximalni cinny vykon alternatoru v dobe poruchy
$P_{max3} = 120.07$	[MW]	Maximalni cinny vykon alternatoru v dobe odpojeni poruchy
$C_{asPoruchy} = 1$	[s]	Cas vzniku poruchy
$C_{asVypnuti} = 1.9$	[s]	Cas vypnuti poruchy
$T_{buz} = 0.3$	[s]	Casova konstanta regulace buzeni
$P_{t0} = 135$	[MW]	Pocatecni vykon turbiny
$T_{tur} = 10$	[s]	Casova konstanta regulace turbiny
$\omega_S = 100\pi$	[rad/s]	Synchronni uhlova rychlost soustroji
$T_m = 5$	[s]	Mechanicka casova konstanta soustroji
$S_{ng} = 250$	[MVA]	Jmenovity zdanlivy vykon alternatoru

## Model



## Data

```
*: Jednoduchý model stability alternátoru
*SYSTEM;
```

```
Pmax1=225.03;  :: [MW] Maximalni cinny vykon alternatoru v predporuchovem stavu
Pmax2=105.23;  :: [MW] Maximalni cinny vykon alternatoru v dobe poruchy
Pmax3=120.07;  :: [MW] Maximalni cinny vykon alternatoru v dobe odpojeni poruchy
```

```
CasPoruchy=1;    :: [s] Cas vzniku poruchy
CasVypnuti=1.9;  :: [s] Cas vypnuti poruchy
```

```
: Regulace buzeni
```

```

Tbuz=0.3;          :: [s] Casova konstanta regulace buzeni
RegBuzeni=1+(TIME>=CasPoruchy)*(1-EXP(-(TIME-CasPoruchy)/Tbuz))*1.5;
:: [-] Pomerna hodnota regulace buzeni

: Vykon alternatoru
Pg=RegBuzeni*
  ((TIME< CasPoruchy)*Pmax1*SIN(Theta)+
  (TIME>=CasPoruchy)*(TIME<CasVypnuti)*Pmax2*SIN(Theta)+
  (TIME>=CasVypnuti)*Pmax3*SIN(Theta));
:: [MW] Cinny elektricky vykon alternatoru

: Vykon turbiny
Pt0=135;          :: [MW] Pocatecni vykon turbiny
Ttur=10;          :: [s] Casova konstanta regulace turbiny
RegTurbiny=1-(TIME>=CasPoruchy)*(1-EXP(-(TIME-CasPoruchy)/Ttur))*0.9; :: [-] Pomerna hodnota
regulace turbiny
Pt=RegTurbiny*Pt0; :: [MW] Mechanicky vykon turbiny



OmegaS=100pi;    ::[rad/s] Synchronni uhlova rychlost soustroji
Tm=5;            ::[s]      Mechanicka casova konstanta soustroji
Sng=250;         ::[MVA]   Jmenovity zdanlivy vykon alternatoru
Jmot=Tm*Sng/(OmegaS*OmegaS);



: Pohybove rovnice soustroji
Mt > J dOmega = -Pt/OmegaS;
Mg > J dOmega = Pg/OmegaS;
Setrvacnost > C dOmega = Jmot;
Integrator > @Int dOmega,Theta;

DeltaOmega=dOmega; :: [rad/s] Relativni uhlova rychlost soustroji
ZateznyUhel=Theta; :: [rad] Zatezny uhel alternatoru

*TR;
TR 0 2;
PRINT(1001) ZateznyUhel, DeltaOmega, Pg, Pt,
  RegBuzeni, RegTurbiny;
INIT Theta=0.64341;
RUN;
*END;

```





**Origin**

Karel Nohac, KEE, FEL, ZCU v Plzni

**Last Update**

December 17, 2018